Глава 6. МОДЕЛИРОВАНИЕ СМЕШАННЫХ АНАЛОГО-ЦИФРОВЫХ УСТРОЙСТВ 





6.1. Основные понятия 





Первоначально программа PSpice была предназначена для моделирования чисто аналоговых устройств. В версии PSpice 3.05 появилась возможность логического моделирования цифровых устройств и интерфейса между аналоговыми и цифровыми компонентами. Однако при этом аналоговые и цифровые блоки должны были располагаться последовательно. Сначала, например, моделируется аналоговый блок, результаты расчетов оцифровываются и записываются в файл, который передается в отдельную программу логического моделирования цифрового устройства и наоборот. Начиная с версии 4 программа PSpice обеспечивает принципиально новую возможность моделирования произвольных смешанных аналого-цифровых цепей с обратными связями, которые, в частности, могут состоять только из цифровых устройств и не содержать аналоговых блоков. Обычно смешанные цепи моделируются в режиме .TRAN (расчет переходных процессов), однако другие режимы также доступны. В режиме .DC задержки сигналов в цифровых блоках игнорируются и рассчитываются логические уровни выходов цифровых устройств в стационарном режиме. В режимах .AC, .NOISE, .TF и .SENS цифровые компоненты не участвуют в анализе малосигнальных частотных характеристик цепи, лишь для аналоговых частей аналого-цифровых и цифро-аналоговых интерфейсов составляются линеаризированные схемы замещения их входных и выходных комплексных сопротивлений. Директивы программы PSpice описаны в разд. 3.3, здесь же обсудим специфику моделирования цифро-аналоговых устройств. 


Реальные цифровые ИС в программе PSpice представлены в виде примитивов Uxxx, отражающих их функционирование на логическом уровне, и двух аналого-цифровых и цифро-аналоговых интерфейсов А/Ц и Ц/А, отображающих их входные и выходные каскады (рис. 6.1). В задании на моделирование указываются только примитивы цифровых устройств Uxxx. Если при этом цифровые ИС соединяются непосредственно друг с другом, то блоки интерфейсов во внимание не принимаются. Если же ко входу или выходу ИС подключен аналоговый компонент, то необходимо включить соответствующий интерфейс. В принципе они могут быть вручную включены в описание задания на моделирование (в файле .cir) как отдельные компоненты, однако программа PSpice делает это автоматически. 

Рис. 6.1. Модель цифровой ИС


Итак, смешанные аналого-цифровые цепи состоят из компонентов трех типов: 1) аналоговые компоненты; 2) устройства сопряжения аналоговых и цифровых компонентов (устройства интерфейса); 3) цифровые компоненты (примитивы). 


Соответственно различают три типа узлов: 1) аналоговые узлы, к которым подключены только аналоговые устройства, 2) цифровые узлы, к которым подключены только цифровые устройства, 3) узлы интерфейса, к которым подключена комбинация аналоговых и цифровых устройств. Программа PSpice автоматически расщепляет каждый узел интерфейса на два узла ( чисто аналоговый и чисто цифровой ( и включает между ними макромодель аналого-цифрового или цифро-аналогового интерфейса. Кроме того, к моделям интерфейсов автоматически подключается источник питания цифровых схем. 


Логические уровни цифровых узлов принимают одно из пяти значений [2]: 


1 ( высокий уровень;


0 ( низкий уровень;


R ( положительный фронт (Raise, переход из состояния “0” в “1”);


F ( отрицательный фронт (Fall, переход из состояния “1” в “0”);


X ( неопределенное состояние (может принимать значение “0”, “1”, промежуточное или нестабильное состояние). 


При вычислении логических уровней узлов, к которым подключено несколько цифровых компонентов, принимаются во внимание выходные сопротивления источников сигналов. 


Приведем далее описания устройств интерфейса и цифровых компонентов. Примеры библиотек отечественных цифровых и аналого-цифровых ИС приведены в Приложении 5. 





6.2. Устройства интерфейса 





Устройства интерфейса включаются между аналоговыми и цифровыми компонентами и выполняют две функции. Во-первых, с их помощью при моделировании электрических процессов в аналоговой части цепи задаются схемы замещения входных и выходных каскадов цифровых компонентов, соединенных непосредственно с аналоговыми компонентами. Во-вторых, они обеспечивают преобразование электрического напряжения в логический уровень и наоборот, чтобы обеспечить обмен данными между подпрограммами моделирования электрических процессов в аналоговой части цепи и логического моделирования цифровой части. Они подразделяются на устройства передачи данных от аналоговых на вход цифровых компонентов, называемые интерфейс А/Ц (Digital Output), и на устройства передачи данных от цифровых на вход аналоговых компонентов, называемые интерфейс Ц/А (Digital Input). 


Если аналоговые и цифровые компоненты взаимодействуют в процессе моделирования, устройства интерфейса включаются в схему замещения цепи автоматически, когда они соединяются друг с другом. Пользователь при этом не включает устройство сопряжения в файл описания цепи с расширением .cir. Для обеспечения такого режима предварительно в библиотеки цифровых компонентов включаются ассоциируемые с каждым компонентом модели устройств интерфейса, оформленные в виде макромоделей. При расщеплении узла интерфейса для автоматического включения устройства интерфейса программа PSpice создает новый цифровой узел. Перечень этих узлов, созданных программой, помещается в выходном файле с расширением .out в разделе Generated AtoD and DtoA. Заметим, что узел интерфейса характеризуется электрическим напряжением, а дополнительный цифровой узел ( логическим состоянием. Пользователь имеет право непосредственно включить в описание цепи устройства интерфейса А/Ц и Ц/А, при этом программа новые устройства включать уже не будет. 


Рассмотрим в качестве примера смешанную цепь на рис. 6.2, а. Она состоит из двух цифровых компонентов ( генератора цифрового сигнала U1 и логического вентиля 2И(НЕ, представленного в виде макромодели X1. На один вход вентиля X1 подключается генератор аналогового сигнала VSIN, на другой ( генератор цифрового сигнала U1, а к выходу ( аналоговая RС-цепь. Таким образом, здесь имеются два узла интерфейса 1 и 2. К узлу 1 подключены аналоговый компонент VSIN и вход цифрового компонента X1, поэтому между этими компонентами программа включит в схему замещения интерфейс А/Ц и создаст дополнительный цифровой узел 1$AtoD. Аналогично между выходом цифрового компонента X1 и аналоговой RС-цепью будет включен интерфейс Ц/А и создан дополнительный цифровой узел 3$DtoA, как показано на рис. 6.2, б. Имена дополнительных цифровых узлов составляются по следующему правилу: в начале их имени повторяется имя узла интерфейса, к нему добавляется символ $ и затем суффикс AtoD или DtoA в зависимости от типа интерфейса (имя узла на выходе А/Ц приобретает суффикс AtoD, на входе Ц/А ( DtoA). Если к одному узлу интерфейса подключается не один, а несколько однотипных компонентов, то образуются дополнительные цифровые узлы, в конце имен которых добавляются цифры 2, 3 ... Кроме того, в схеме замещения на рис. 6.2, б к интерфейсу Ц/А автоматически подключается источник питания через глобальный узел $G_POS. 

Рис. 6.2. Пример смешанной аналого-цифровой цепи (а) и ее схемы замещения (б)


Отметим, что схема замещения на рис. 6.2, б содержит аналоговые узлы 1, 3, $G_POS и цифровые узлы 1$AtoD, 2, 3$DtoA. Потенциалы аналоговых узлов и логические состояния цифровых узлов можно вывести на печать, например, по директиве 





.PRINT   TRAN   V(1)   D(2)   V(3) 





Итак, каждому реальному цифровому компоненту в программе PSpice ставятся в соответствие: 


1) два устройства интерфейса для сопряжения с аналоговыми устройствами, которые могут подключаться к его входу и выходу, названные нами для краткости интерфейсами Ц/А и А/Ц; они осуществляют обмен данными между подпрограммами моделирования аналоговых и цифровых устройств программы PSpice; 


2) модели вход/выход, отображающие его входные и выходные комплексные сопротивления; 


3) модели динамики, учитывающие запаздывания сигналов. Модели цифровых компонентов и ассоциируемые с ними сопутствующие модели помещаются в специальные библиотеки. 





6.2.1. Аналого-цифровой интерфейс 





Аналого-цифровые интерфейсы предназначены для преобразования аналогового напряжения в логический уровень. Они имитируют входные каскады цифровых ИС. Их схема замещения показана на рис. 6.3, а. Аналого-цифровые интерфейсы не обязательно включать в схему устройства вручную, так как программа PSpice автоматически расщепляет узел интерфейса и включает устройства интерфейса между аналоговым узлом и входным узлом цифрового устройства. Информация о логическом уровне сигнала на выходе интерфейса А/Ц может направляться двояко: 


( в модуль логического моделирования программы PSpice; 


( в файл (позднее этот файл может быть просмотрен визуально или использован в качестве источника цифрового сигнала при логическом моделировании в следующем сеансе). 

Рис. 6.3. Аналого-цифровой (а) и цифро-аналоговый (б) интерфейсы





При взаимодействии аналоговых и цифровых компонентов в процессе моделирования описание интерфейса А/Ц (digital output: analog-to-digital) имеет формат 





Oxxx <+узел интерфейса> <(узел интерфейса> <имя модели А/Ц> 


+ DGTLNET=<имя цифрового узла> <имя модели вход/выход> 





При записи логических уровней в файл описание интерфейса А/Ц имеет формат 





Oxxx <+узел интерфейса> <(узел интерфейса> <имя модели А/Ц> 


+ [SIGNAME=<имя цифрового сигнала>] 





Приведем примеры: 





O12   ANALOG_NODE   DIGITAL_GND   DOUT_133 


+  DGTLNET=DIG_NODE   IO_STD 


OVCO   17   0   TO_TTL   SIGNAME=VCO_OUT; передача данных в файл 





Модель аналого-цифрового интерфейса описывается предложением 





.MODEL <имя модели А/Ц> DOUTPUT [<параметры модели>] 





Параметры модели аналого-цифрового интерфейса приведны в табл. 6.1.





Таблица 6.1





Идентификатор �
Параметр �
Значение по умолчанию�
Единица измерения �
�
RLOAD �
Сопротивление нагрузки �
1/GMIN�
Ом �
�
CLOAD �
Емкость нагрузки �
0 �
Ф �
�
CHGONLY �
Флаг преобразования: 0 ( преобразование на каждом шаге по времени; 1 ( преобразование при наличии изменения входного напряжения (только при записи в файл)�
�
 �
�
S0NAME �
Имя логического состояния “0” �
�
�
�
S0VLO �
Нижний уровень напряжения логического “0” �
�
В �
�
S0VHI �
Верхний уровень напряжения логического “0”  �
�
В�
�
S1NAME �
Имя логического состояния “1” �
�
�
�
S1VLO �
Нижний уровень напряжения логической “1”  �
�
В�
�
S1VHI �
Верхний уровень напряжения логической “1”�
�
В�
�
.    .    .    .    .   �
.    .    .    .    .    .    .    .    .    .    .�
.    .    .    .    .�
.    .    .    �
�
S19NAME �
Имя логического состояния “19”  �
�
�
�
S19VLO �
Нижний уровень напряжения логического состояния “19”   �
�
В�
�
S19VHI �
Верхний уровень напряжения логического состояния “19”�
�
В�
�
SXNAME �
Имя логического состояния, когда напряжение на узле интерфейса находится вне заданных границ�
?�
�
�
FILE �
Имя файла цифрового сигнала (только  при записи в файл)�
�
 �
�
FORMAT �
Код формата входного файла цифровых сигналов (только при записи в файл)�
1�
�
�
TIMESTEP �
Интервал дискретизации по времени при  с записи в файл �
10� EMBED Equation.2  ����
c�
�
TIMESCALE �
Масштабный коэффициент при расчете интервала времени TIMESTEP (только при записи в файл) �
1�
�
�



Каждому i-му логическому состоянию соответствует определенный диапазон напряжений SiVL0...SiVHI. До тех пор пока входное напряжение интерфейса А/Ц Vвх не выходит за его границы, логическое состояние на выходе интерфейса А/Ц не изменяется. В противном случае входное напряжение будет сравниваться с пороговыми уровнями, начиная с S0VLO, пока оно не попадет в какой-нибудь интервал. Если оно не попадает ни в один интервал напряжений, логическому состоянию интерфейса А/Ц по умолчанию  присваивается символ “?” (его можно изменить с помощью параметра SXNAME). 


При взаимодействии с модулем логического моделирования программы PSpice имена логических состояний должны быть “0”, “1”, “X”, “R”, “F” или “Z” (однако состояние “Z” обычно не используется, так как состояние высокого импеданса не определяет уровень напряжения). 


Параметры TIMESCALE, FILE, CHGONLY и FORMAT используются только при записи цифрового сигнала в файл. Параметр TIMESTEP определяет точность установления момента времени изменения логического состояния. При расчете переходных процессов в аналоговых цепях время интегрирования не может превышать значение TIMESTEP. По умолчанию этот параметр принимает меньшее из значений 1 нс или 1/DIGFREQ (параметр DIGFREQ устанавливается по директиве .OPTIONS). Моменты времени, в которые записываются значения выборок сигнала, равны целым значениям величин 





k = TIMESCALE(TIME/TIMESTEP,





где TIME ( текущее время моделирования, k ( номер отсчета. По умолчанию TIMESCALE=1. Для экономии машинного времени следует назначать значение параметра TIMESTEP, исходя из скорости изменения напряжения. 


При записи дискретизированного сигнала в файл его имя указывается в опции 





SIGNAME=<имя файла> 





Если эта опция опущена, то в качестве имени файла будет принято имя интерфейса А/Ц Oxxx без первого символа O. 


Логическое состояние интерфейса А/Ц можно просмотреть с помощью программы Probe, указывая в качестве имени переменной B(Oxxx), см. комментарий к интерфейсу Ц/А. 


Моделирование входной цепи реального цифрового компонента с помощью линейной RС-цепи, как показано на рис. 6.3, а, не всегда обеспечивает достаточную точность расчетов. Поэтому пользователь должен иметь в своем распоряжении несколько моделей входных цепей логических компонентов, оформляя их в виде макромоделей. В качестве примера на рис. 6.4 представлена нелинейная модель входной цепи логических элементов ТТЛ, к которой подключено стандартное устройство сопряжения типа Oxxx. 

Рис. 6.4. Нелинейная модель входной цепи цифрового компонента





6.2.2. Цифро-аналоговый интерфейс 





Цифро-аналоговый интерфейс предназначен для преобразования логического уровня выходных сигналов цифровых компонентов (“0”, “1”, “X”, “R”, “F” или “Z”) в аналоговое напряжение, как показано на рис. 6.3, б. Эти устройства включают на входе аналоговых компонентов. Аналоговое напряжение образуется с помощью источника опорного напряжения и делителя на резисторах, сопротивления которых изменяются программно в соответствии с логическим уровнем цифрового сигнала. Информация о логическом уровне сигнала может быть получена из двух источников: 


( в процессе расчета временной диаграммы логических устройств с помощью программы PSpice; 


( из файла, который создан предварительно вручную или образован в результате моделирования в предыдущем сеансе работы с программой. 


В связи с этим цифро-аналоговый интерфейс описывается двояко. При взаимодействии аналоговых и цифровых компонентов в процессе моделирования его описание имеет вид (digital input: digital-to-analog): 





Nxxx <узел интерфейса> <(узел источника опорного напряжения> 


+ <+узел источника опорного напряжения> <имя модели Ц/А> 


+ DGTLNET=<имя цифрового узла> <имя модели вход/выход> 


+ [IS=<начальное состояние>] 





При управлении из файла описание цифро-аналогового интерфейса имеет формат 





Nxxx <узел интерфейса> <(узел источника опорного напряжения> 


+ <+узел источника опорного напряжения> <имя модели Ц/А> 


+ [SIGNAME=<имя цифрового сигнала>] 


+ [IS=<начальное состояние>] 





Например





N1   ANALOG  DIGITAL_GND  U_REF DIN_133  DGTLNET=13 IO_STD 


N2   7   15   16  FROM_TTL	; Данные передаются из файла, имя которого


					; указано в модели FROM_TTL 





Модель цифро-аналогового интерфейса описывается предложением 





.MODEL <имя модели Ц/А> DINPUT [<параметры модели>] 





Параметры модели цифро-аналогового интерфейса приведены в табл. 6.2.





Таблица 6.2





Идентификатор  �
Параметр �
Значение по умолчанию �
Единица измерения�
�
CLO �
Емкость между выходным узлом и “землей” �
0 �
Ф�
�
CHI�
Емкость между выходным узлом и источником питания �
0 �
Ф�
�
SONAME �
Имя состояния логический “0”�
�
�
�
SOTSW �
Время переключения в состояние “0” �
 �
с �
�
SORLO �
Сопротивление между выходным узлом и "землей" в состоянии “0”�
�
Ом�
�
SORHI �
Сопротивление между выходным узлом и источником питания в состоянии “0”  �
�
Ом�
�
S1NAME �
Имя состояния логическая “1”�
�
�
�
S1TSW �
Время переключения в состояние “1” �
�
с �
�
S1RLO �
Сопротивление между выходным узлом и “землей” в состоянии “1” �
�
Ом�
�
S1RHI �
Сопротивление между выходным узлом и источником питания в состоянии “1” �
�
Ом�
�
S19NAME �
Имя логического состояния “19”�
�
�
�
S19TSW �
Время переключения в состояние “19”�
�
с �
�
S19RLO �
Сопротивление между выходным узлом и “землей” в состоянии “19”�
�
 Ом�
�
S19RHI �
Сопротивление между выходным узлом и источником питания в состоянии “19”�
�
Ом�
�
FILE �
Имя файла с цифровыми сигналами  (только при чтении  из файла) �
�
�
�
FORMAT �
Код формата входного файла (только при чтении  из файла) �
1�
�
�
TIMESTEP �
Интервал времени между соседними отсчетами цифрового сигнала в файле  (только при чтении из файла)�
10� EMBED Equation.2  ��� �
с�
�



Опция 





DGTLNET=<имя цифрового узла> <имя модели вход/выход> 





указывает имя цифрового узла, к которому подсоединен интерфейс Ц/А, и имя модели входного/выходного сопротивления соответствующего цифрового компонента (см. ниже). 


Начальное логическое состояние управляющего цифрового узла в момент времени t=0 определяется режимом схемы по постоянному току. Изменение этого состояния при необходимости производится с помощью необязательной опции 





IS=<начальное состояние> 





Параметры модели FILE, FORMAT и TIMESTEP используются только при управлении из файла. При этом управление производится цифровым сигналом, имя которого указано в необязательной опции 





SIGNAME=<имя цифрового сигнала> 





В ее отсутствие управление осуществляется сигналом, имя которого образовано символами xxx, находящимися справа от префикса N имени цифро-аналогового интерфейса Nxxx. 


На рис. 6.3, б изображена схема замещения выходного каскада цифровых ИС. Сопротивления резисторов изменяются в соответствии с логическим уровнем управляющего цифрового узла. При его изменении сопротивления резисторов плавно изменяются по экспоненциальному закону с постоянной времени, которая определяется длительностью переключения, указанной в модели DINPUT. Эти сопротивления рассчитываются по формулам 





RLO=R� EMBED Equation.2  ���E� EMBED Equation.2  ���/(E� EMBED Equation.2  ��� ( U� EMBED Equation.2  ���),   RHI=R� EMBED Equation.2  ���E� EMBED Equation.2  ���/U� EMBED Equation.2  ���, 





где R� EMBED Equation.2  ��� и U� EMBED Equation.2  ��� - выходное сопротивление и уровень выходного напряжения в данном логическом состоянии, E� EMBED Equation.2  ��� - напряжение источника питания. 


К управляющему цифровому узлу не должны подключаться какие-либо аналоговые компоненты, так как в противном случае между ними автоматически будет включен аналого-цифровой интерфейс и цифровой узел отсоединится. 


В процессе моделирования логические состояния управляющего цифрового узла получают имена “0”, “1”, “X”, “R”, “F” и “Z”. Моделирование прерывается, если хотя бы одно из этих состояний не указано в перечне спецификаций SnNAME, SnTSW, SnRLO и SnRHI. 


Логическое состояние входного цифрового сигнала можно просмотреть с помощью программы Probe на аналоговом экране, указывая в качестве имени переменной B(Nxxx). Значение B(Nxxx) равно 0.0, если текущее состояние есть S0NAME, равно 1.0 в состоянии S1NAME и т.д. до 19.0. Однако логические состояния входных цифровых узлов удобнее выводить на цифровой экран программы Probe. 





6.2.3. Модель вход/выход 





Модели вход/выход, ассоциируемые с каждым цифровым компонентом, имеют тип UIO и задаются по формату 





.MODEL <имя модели вход/выход> UIO [<параметры модели>] 





Параметры модели вход/выход приведены в табл. 6.3. 





Таблица 6.3





Идентификатор параметра�
Параметр �
Значение по умолчанию �
Единица измерения �
�
INLD �
Входная емкость �
0 �
Ф �
�
OUTLD �
Выходная емкость �
0 �
Ф �
�
DRVH �
Выходное сопротивление высокого уровня �
50 �
Ом�
�
DRVL �
Выходное сопротивление низкого уровня �
50�
Ом�
�
DRVZ �
Выходное сопротивление утечки цепи, моделируемой как цепь  хранения заряда�
250 кОм�
Ом�
�
INR�
Входное сопротивление утечки цепи, моделируемой как цепь хранения заряда�
30 кОм�
Ом�
�
TSTOREMN�
Минимальное время сохранения заряда цепи, моделируемой как цепь  хранения заряда�
1 мс�
с�
�
AtoD1 �
Имя макромодели интерфейса А/Ц первого уровня �
AtoDDefault�
�
�
DtoA1 �
Имя макромодели интерфейса Ц/А первого уровня �
DtoADefault�
�
�
AtoD2 �
Имя макромодели интерфейса А/Ц второго уровня �
AtoDDefault�
�
�
DtoA2 �
Имя макромодели интерфейса Ц/А второго уровня �
DtoADefault�
�
�
AtoD3 �
Имя макромодели интерфейса А/Ц третьего  уровня �
AtoDDefault�
�
�
DtoA3 �
Имя макромодели интерфейса Ц/А третьего  уровня�
 DtoADefault�
�
�
AtoD4 �
Имя макромодели интерфейса А/Ц четвертого уровня �
AtoDDefault�
�
�
DtoA4 �
Имя макромодели интерфейса Ц/А четвертого уровня �
DtoADefault�
�
�
TSWLH1 �
Время переключения 0� EMBED Equation.2  ���1 для DtoA1 �
0 �
с �
�
TSWLH2 �
Время переключения 0� EMBED Equation.2  ���1 для DtoA2 �
0 �
с �
�
TSWLH3 �
Время переключения 0� EMBED Equation.2  ���1 для DtoA3 �
0 �
с �
�
TSWLH4 �
Время переключения 0� EMBED Equation.2  ���1 для DtoA4 �
0 �
с �
�
TSWHL1 �
Время переключения 1� EMBED Equation.2  ���0 для DtoA1 �
0 �
с �
�
TSWHL2 �
Время переключения 1� EMBED Equation.2  ���0 для DtoA2 �
0 �
с �
�
TSWHL3 �
Время переключения 1� EMBED Equation.2  ���0 для DtoA3 �
0 �
с �
�
TSWHL4 �
Время переключения 1� EMBED Equation.2  ���0 для DtoA4 �
0 �
с �
�
TPWRT�
Пороговое значение длительности импульса�
Равно минимальной задержке�
с�
�
DIGPOWER�
Имя макромодели источника питания�
DIGIFPWR�
�
�



Входная и выходная емкости INLD, OUTLD принимаются во внимание при расчете времен задержки. Выходные сопротивления цифровых устройств задаются параметрами DRVH, DRVL модели вход/выход UIO (рис. 6.3, а). Выходное сопротивление компонента, находящегося в состоянии “1”, обозначается как DRVH, в состоянии “0” ( DRVL. В программе PSpice выходные сопротивления компонентов принимают значения в диапазоне от DIGDRVF (Forcing strength) до DIGDRVZ (Z strength), который в логарифмическом масштабе разбивается на 64 уровня (максимальному сопротивлению DIGDRVZ присваивают код 0, а минимальному DIGDRVF ( код 63). По умолчанию DIGDRVF=2 Ом, DIGDRVZ=20 кОм; их значения переназначаются по директиве .OPTIONS. В конфликтных ситуациях, когда к одному узлу подключаются вентили с разными выходными сопротивлениями, логический уровень узла устанавливается вентилем с минимальным выходным сопротивлением,  код которого больше кодов остальных сопротивлений в заданное число раз. Это отношение кодов сопротивлений задается параметром DIGOVRDRV директивы .OPTIONS, который по умолчанию равен 3. Когда имеется несколько вентилей с близкими выходными сопротивлениями и разными логическими уровнями, узлу присваивается неопределенное состояние X. 


Времена переключения выходных каскадов цифровых ИС задаются параметрами TSWLHn, TSWHLn (трудности их определения по справочным данным заключаются в том, что обычно приводятся значения общего времени переключения всей ИС). 


Макромодели интерфейсов составляются пользователями и включаются в библиотечный файл. Эти модели отражают характер входных/выходных сопротивлений цифровых компонентов с разной степенью подробности. Модели, имеющиеся в стандартной библиотеке интерфейсов программы PSpice, приведены ниже:





IO_LEVEL�
 Определение �
�
0�
Текущее значение параметра DIGIOLVL  директивы .OPTIONS (по умолчанию равно 1)�
�
1 �
Основная (простейшая) модель, имеющая логические состояния 0, 1, X, R и F (AtoD1/DtoA1) �
�
2 �
Основная (простейшая) модель без промежуточного состояния X (AtoD2/DtoA2)�
�
3 �
Сложная модель с промежуточным состоянием X (AtoD3/DtoA3)  �
�
4 �
Сложная модель без промежуточных состояний X, R и F (AtoD4/DtoA4)�
�



Сложные модели точнее имитируют нелинейности входных сопротивлений цифровых ИС, однако требуют больших вычислительных затрат. 


Имена макромоделей интерфейсов указываются с помощью параметров AtoD1, DtoA1, ..., AtoD4, DtoA4. Выбор уровня модели интерфейса для каждого конкретного цифрового устройства производится с помощью параметра IO_LEVEL (см. ниже). 


Схемы замещения простейших интерфейсов первого уровня показаны на рис. 6.3. В текстовом виде они записываются следующим образом. 


Макромодель стандартного ТТЛ-интерфейса А/Ц первого уровня имеет вид: 





.subckt   AtoD_STD   A   D   DPWR   DGND 


+ params: CAPACITANCE=0 


* 


O0   A   DGND   DO74   DGTLNET=D   IO_STD 


C1   A   DGND {CAPACITANCE=0.1pF} 


.ends 


*


.model DO74 doutput ( 


+ s0name="X" 	s0vlo=0.8 	s0vhi=2.0 


+ s1name="0" 	s1vlo=-1.5 	s1vhi=0.8 


+ s2name="R"	s2vlo=0.8 	s2vhi=1.4 


+ s3name="R"	s3vlo=1.3 	s3vhi=2.0 


+ s4name="X" 	s4vlo=0.8 	s4vhi=2.0 


+ s5name="1" 	s5vlo=2.0 	s5vhi=7.0 


+ s6name="F" 	s6vlo=1.3 	s6vhi=2.0 


+ s7name="F" 	s7vlo=0.8 	s7vhi=1.4 ) 


*


.model IO_STD uio (drvh=96.4 drvl=104 


+ AtoD1="AtoD_STD" 	AtoD2="AtoD_STD_NX" 


+ AtoD3="AtoD_STD_E" 	AtoD4="AtoD_STD_NX_E" 


+ DtoA1="DtoA_STD" 	DtoA2="DtoA_STD_NX" 


+ DtoA3="DtoA_STD_E" 	DtoA4="DtoA_STD_NX_E" 


+ tswhl1=1.373ns 		tswlh1=3.382ns 


+ tswhl2=1.346ns 		tswlh2=3.424ns 


+ tswhl3=1.511ns 		tswlh3=3.517ns 


+ tswhl4=1.487ns 		tswlh4=3.564ns ) 





Макромодель стандартного ТТЛ-интерфейса Ц/А первого уровня имеет вид: 





.subckt    DtoA_STD    D    A    DPWR    DGND 


+ params: DRVL=0   DRVH=0   CAPACITANCE=0 


* 


N1   A   DGND   DPWR   DIN74   DGTLNET=D IO_STD 


C1   A   DGND  {CAPACITANCE=0.1pF} 


.ends 


*


.model DIN74 dinput ( 


+ s0name="0"   s0tsw=3.5ns    s0rlo=7.13    s0rhi=389 ;   7ohm, 0.09v 


+ s1name="1"   s1tsw=5.5ns    s1rlo=467     s1rhi=200 ;   140ohm, 3.5v 


+ s2name="X"   s2tsw=3.5ns   s2rlo=42.9    s2rhi=116 ;  31.3ohm, 1.35v 


+ s3name="R"   s3tsw=3.5ns   s3rlo=42.9    s3rhi=116 ;  31.3ohm, 1.35v 


+ s4name="F"   s4tsw=3.5ns   s4rlo=42.9    s4rhi=116 ;  31.3ohm, 1.35v 


+ s5name="Z"   s5tsw=3.5ns   s5rlo=200K  s5rhi=200K ) 





Более сложная модель интерфейса А/Ц первого уровня имеет схему замещения, показанную на рис. 6.4. Ее текстовое описание имеет вид: 





.subckt    AtoD_STD_E   A   D   DPWR DGND 


+ params: CAPACITANCE=0 


* 


O0   A   DGND   DO74   DGTLNET=D   IO_STD 


C1   A   DGND {CAPACITANCE=0.1pF} 


D0   DGND   a   D74CLMP 


D1   1   2   D74 


D2   2   DGND   D74 


R1   DPWR   3   4k 


Q1   1   3   A   0   Q74; подложку соединить с DGND 


.ends 


.model   D74   D (IS=1e-16 RS=25 CJO=2pf) 


.model   D74CLMP   D (IS=1e-15 RS=2 CJO=2pf) 


.model   Q74   NPN (ISE=1e-16 ISC=4e-16 BF=49 BR=.03 CJE=1pf 


+ CJC=.5pf CJS=3pf VJE=0.9v VJC=0.8v VJS=0.7v MIE=0.5 


+ MJC=0.33  MJS=0.33 TF=0.2ns TR=10ns RB=50 RC=20) 








6.2.4. Устройства питания 





В программе PSpice принято, что макромодель источника питания имеет имя DIGIFPWR и на нее автоматически делается ссылка при наличии в схеме макромоделей аналого-цифровых интерфейсов. Конкретное содержание макромодели источника питания определяется пользователем, а ее текст помещается в библиотеку цифровых устройств. Приведем в качестве примера макромодель источника  напряжения, принимающего по умолчанию значение 5 В: 





.subckt   DIGIFPWR   AGND 


+ optional: DPWR=$G_DPWR   DGND=$G_DGND 


+ params: VOLTAGE=5.0v  REFERENCE=0v 


VDPWR   DPWR   DGND {VOLTAGE} 


R1   DPWR   AGND   1MEG 


VDGND   DGND   AGND {REFERENCE} 


R2   DGND   AGND 1MEG 


.ends 





Здесь узлам цифрового питания и цифровой “земли” присвоены глобальные имена $G_DPWR, $G_DGND (при желании их можно изменить здесь и одновременно в моделях всех цифровых устройств). Узел AGND ( общий узел источника питания, который при включении его в общую схему соединяется с узлом аналоговой “земли” 0. Для создания макромодели источника другого напряжения необходимо в задании на моделирование (в файле .cir) включить предложение вызова макромодели источника питания, задав с помощью параметра VOLTAGE нужное значение напряжения питания, и указать после номера узла “земли” 0 имена (номера) узлов его выводов, например: 





XMYPOWER    0   MY_PWR   MY_GND   DIGIFPWR 


+  params: VOLTAGE=9.0v 





Здесь вызывается источник напряжения 9 В, выводы которого имеют имена MY_PWR и MY_GND (эти имена указываются в моделях примитивов цифровых устройств, подключаемых к этому источнику, см. ниже). 


Усложнив макромодель, можно создать источник нескольких напряжений. 





6.3. Генераторы цифровых сигналов 





Генераторы цифровых сигналов можно задать тремя способами. 


1. Определение формы цифрового сигнала в задании на моделирование по формату  (устройства STIM)





Uxxx   STIM(<количество сигналов>,<формат>) 


+ <+узел источника питания> <(узел источника питания> 


+ <список узлов>* 


+ <имя модели вход/выход> 


+ [IO_LEVEL=<номер макромодели интерфейса вход/выход>]


+ [STIMULUS=<имя воздействия>]


+ [TIMESTEP=<шаг по времени>] 


+ <команды описания формы сигнала>* 





Переменная <количество сигналов> определяет количество выходов генератора, равное количеству генерируемых разных цифровых сигналов. 


Переменная <формат> ( это спецификация формата переменной <данные>, в котором представлены логические уровни сигналов генератора. Эта переменная представляет собой последовательность цифр, общее число которых равно значению переменной <количество сигналов>. Каждая цифра принимает значения 1, 3 или 4, что означает двоичную, восьмеричную и шестнадцатеричную систему счисления. 


Подключение источника питания задается номерами узлов <+узел источника питания>, <(узел источника питания>. 


Замечание. В PSpice 4 нельзя было создать несколько источников питания, поэтому номера узлов их подключения не указывались при описании генераторов цифровых сигналов и других цифровых  устройств. Начиная с версии Design Center 5.0 появилась такая возможность, в связи с этим был изменен формат описания цифровых компонентов ( добавлены номера узлов подключения источников питания <+узел источника питания> <(узел источника питания>. Для  преобразования библиотек моделей цифровых компонентов PSpice 4 в новый формат предназначена специальная программа digconv.exe.


Номера подключения выходов генератора к схеме задаются <списком узлов>. 


Имя модели вход/выход задается параметром <имя модели вход/выход>, как для любого цифрового устройства. 


STIMULUS ( необязательный параметр, обозначающий имя сигнала;


IO_LEVEL ( необязательный параметр для выбора одной из четырех макромоделей интерфейса вход/выход (по умолчанию 0);


TIMESTEP ( необязательный параметр для задания периода квантования (или шага). При задании моментов времени номером шага (имеют суффикс “С”) он умножается на величину шага. По умолчанию устанавливается TIMESTEP=0. Этот параметр не принимается во внимание, если заданы абсолютные значения моментов времени (имеют суффикс “S”).


Параметр <описание формы сигнала> представляет собой произвольную комбинацию одной или нескольких следующих строк: 





<t>,<логический уровень> 


LABEL=<имя метки> 


<t> GOTO <имя метки> <n> TIMES 


<t> GOTO < имя метки > UNTIL GT <данные> 


<t> GOTO < имя метки > UNTIL GE <данные> 


<t> GOTO < имя метки > UNTIL LT <данные> 


<t> GOTO < имя метки > UNTIL LE <данные> 


<t> INCR BY <данные> 


<t> DECR BY <данные> 


REPEAT  FOREVER


REPEAT  <n>  TIMES


ENDREPEAT


FILE=<имя файла>





Каждая цифра переменной <данные> представляет собой логический уровень соответствующего выходного сигнала, который представлен в системе счисления 2m, где m ( соответствующая цифра переменной <формат>. 


Количество узлов в <списке узлов> должно быть равно значению переменной <количество сигналов>. 


Переменная <t> определяет моменты времени, в которые задаются логические уровни сигнала. Если перед значением переменной <t> имеется символ “+”, то эта переменная задает приращение относительно предыдущего момента времени; в противном случае она определяет абсолютное значение относительно начала отсчета времени t=0. Суффикс “S” указывает размерность времени в секундах (допускается суффикс “nS” ( наносекунды и т. п.). Суффикс “С” означает измерение времени в количестве циклов, размер которых определяется параметром TIMESTEP (переменная <шаг по времени>). 


Переменная <данные> состоит из символов “0”, “1”, “X”, “R”, “F” или “Z”, интерпретируемых в заданном формате. 


Переменная <n> задает количество повторяющихся циклов GOTO; значение n=(1 задает бесконечное повторение цикла. 


Переменная <имя метки> используется при организации цикла с помощью оператора перехода GOTO, который передает управление на строку, следующую за оператором LABEL=<имя метки>. 


REPEAT  FOREVER ( начало бесконечно повторяющегося цикла (эквивалентно конструкции REPEAT (1 TIMES).


REPEAT  <n>  TIMES ( повторение n раз.


ENDREPEAT ( конец цикла REPEAT.


FILE ( указание имени файла, в котором находится описание одного или нескольких входных сигналов.





Приведем пример: 





USIGNAL   STIM(2,11)   $G_DPWR   $G_DGND   1   2   IO_STM 


+ TIMESTEP=1ns 


+ 0C  00 ; В момент времени t=0 состояния обоих узлов равны "0" 


+ LABEL=M 


+ 1C   01 ; При t=1 нс состояние узла 1 равно "0, узла 2 - "1" 


+ 2C   11 ; При t=2 нс состояние обоих узлов равны "1" 


+ 3С   GOTO M 3 TIMES; В момент времени t=3 нс переход на


+              ; метку M  и выполняется первая инструкция без 


+               ; дополнительной  задержки. Цикл повторяется 3 раза 





2. Считывание  временной диаграммы цифрового сигнала из  файла (устройства FSTIM).  Устройства FSTIM задаются по формату:





Uxxx   FSTIM(<количество выходов>) 


+ <+узел источника питания> <(узел источника питания> 


+ <список узлов>* 


+ <имя модели вход/выход> 


+ FILE=<имя файла сигналов>


+ [IO_LEVEL=<номер макромодели интерфейса вход/выход>]


+ [STIMULUS=<имя воздействия>]





Параметр <количество выходов> определяет количество узлов, к которым подключаются сигналы. Остальные параметры имеют тот же смысл, что в п. 1. Обратим внимание, что задание с помощью параметра FILE имени файла, в котором описаны сигналы, обязательно. 


Файл цифрового сигнала может быть получен путем редактирования файла результатов моделирования или составлен вручную с помощью любого текстового редактора. 


Файл цифрового сигнала имеет две секции: 


1) заголовок (header), содержащий список имен сигналов;


2) список значений сигнала (transitions), содержащий на одной или более строках моменты времени изменений сигналов и колонки их значений. 


Приведем сначала пример файла цифровых сигналов: 





* Заголовок:    содержит имена сигналов 


CLOCK, RESET, IN1, IN2; имена 4-х сигналов 


*Начало описания сигналов; отделяется от заголовка пустой строкой 


0 	0000 ; двоичные коды сигналов 


10ns 	1100 


20ns 	0101 


30ns 	1110 


40ns 	0111 





Заголовок имеет следующий формат: 





[TIMESCALE=<значение>] 


<имя сигнала 1>...<имя сигнала n>... 


OCT(<3-й бит сигнала>...<1-й бит сигнала>)... 


HEX(<4-й бит сигнала>...<1-й бит сигнала>)... 





Имена сигналов могут разделяться запятыми или пробелами. Они располагаются на одной или нескольких строках, но строки продолжения не имеют в начале знака “+”. Максимальное количество сигналов 255, на одной строке могут размещаться не более 300 символов. Имена сигналов перечисляются в том же порядке, в котором их значения приводятся во временных диаграммах. 


Если перед именем сигнала не указывается название системы счисления, он считается двоичным. Имена сигналов, записанных в восьмеричном коде, предваряются кодом OCT и группируются по 3 сигнала (имена групп сигналов заключаются в круглые скобки). Шестнадцатеричные сигналы имеют код HEX и группируются по 4 сигнала. 


Приведем еще один пример файла сигналов: 





Clock Reset In1 In2 


HEX(Addr7 Addr6 Addr5 Addr4)   HEX(Addr3 Addr2 Addr1 Addr0) 


ReadWrite 


0 		0000 		00 	0 ;  Для наглядности между группами


10n 	1100 		4E 	0 ;  можно включать пробелы 


20n 	0101 		4E 	1 


30n 	1110 		4E 	1 


40n 	0111 		FF 	0 





Здесь в начале приведены имена четырех двоичных сигналов, затем две группы сигналов, задаваемых в шестнадцатеричном виде, и затем еще один бинарный сигнал. В последующем списке значений сигналов даны 7 колонок, соответствующих приведенным в заголовке сигналам. 


Список значений имеет формат: 





<время> <значение сигнала>* 





Список значений отделяется от заголовка пустой строкой. Моменты времени и список значений сигналов должны быть разделены по крайней мере одним пробелом. Для моментов времени изменений сигналов указываются их абсолютные значения (в секундах), например 25ns, 1.2E-9, 5.8, или приращение относительно предыдущего момента времени. Признак приращения ( знак “+” перед значением момента времени, например +5ns. Каждое значение сигнала соответствует одиночному бинарному сигналу или группе сигналов, имена которых помещены в группы с указателями системы счисления OCT или HEX. Общее количество значащих цифр в значениях сигналов должно быть равно общему количеству бинарных сигналов и групп сигналов, помеченных символами OCT и HEX. Разряды данных сигнала могут принимать следующие значения: 





�
Двоичные �
Восьмеричные �
Шестнадцатиричные �
�
Логические/цифровые �
0, 1 �
0...7 �
0...F �
�
Неопределенные �
X �
X �
X �
�
Высокого импеданса �
Z �
Z �
Z �
�
Передний фронт �
R �
R �
R �
�
Задний фронт �
F �
F�
-�
�



Заметим, что задний фронт не может в шестнадцатеричной системе обозначаться как F, потому что этот символ занят под численное значение. 


Приведем примеры генераторов цифровых сигналов типа FSTIM.


1. Пусть имеется файл dig1.stm, в котором описана временная диаграмма сигнала IN1. Тогда в задание на моделирование можно включить следующее описание генератора:





U1  FSTIM(1)  $G_DPWR  $G_DGND


+  IN1  IO_STM  FILE=DIG1.STM





2.  Пусть имеется файл flipflop.stm следующего содержания:





J   K  PRESET  CLEAR  CLOCK





0		0	0	010


10ns	0	0	111


.  .  .  .  .


В задание на моделирование включим описание генератора сигнала





U2  FSTIM(4)  $G_DPWR  $G_DGND


+  CLK  PRE  J  K


+  IO_STM


+  FILE=FLIPFLOP.STM


+  SIGNAMES = CLOCK  PRESET





В этом примере первым двум узлам CLK и PRE соответствуют сигналы CLOCK и PRESET из файла сигналов, это соответствие устанавливается с помощью опции SIGNAMES. Последним узлам J и K  соответствуют одноименные сигналы из файла сигналов, поэтому их имена нет необходимости включать в опцию SIGNAMES. Сигнал  CLEAR в данном примере не используется (но на него можно сослаться в другом генераторе FSTIM).


6.4. Цифровые компоненты 





Цифровые компоненты (примитивы) задаются по формату 





Uxxx <тип> [(<список параметров>*)] 


+ <+узел источника питания> <(узел источника питания> 


+ <список узлов>* 


+ <имя модели динамики> <имя модели вход/выход> 


+ [MNTYMXDLY=<выбор значения задержки>] 


+ [IO_LEVEL=<уровень модели интерфейса>] 





Параметр <тип> указывает тип логического устройства (их перечень приведен ниже, например AND, NOR); в круглых скобках указываются значения одного или более параметров через запятую (например, для схемы И указывается количество входов). После списка узлов подключения логического устройства следуют имена двух моделей. Первая модель описывает динамические свойства устройства, вторая ( характеристики входных и выходных сопротивлений. 


Модели динамики  имеют следующие ключевые слова:





Имя модели динамики�
Тип компонента �
�
UADC�
 Аналого-цифровой преобразователь �
�
UDAC �
Цифро-аналоговый преобразователь �
�
UIO �
Модель входа/выхода цифрового устройства �
�
UGATE �
Стандартный вентиль �
�
UTGATE �
Вентиль с тремя состояниями �
�
UBTG�
Двунаправленный переключающий вентиль�
�
UEFF �
Триггер с динамическим управлением �
�
UGFF �
Триггер с потенциальным управлением �
�
UDLY �
Цифровая линия задержки �
�
UPLD�
Программируемые логические матрицы�
�
UROM �
Постоянное запоминающее устройство �
�
URAM�
Оперативное запоминающее устройство�
�



Модели вход/выход имеют ключевое слово UIO (разд. 6.2.3). 


Параметр MNTYMXDLY позволяет конкретному индивидуальному устройству назначить минимальное, типичное или максимальное значение времени задержки, указанное в спецификации модели его динамики: 


0 ( значение задержки, заданное параметром DIGMNTYMX  директивы .OPTIONS (по умолчанию параметр равен 2);


1 ( минимальное значение;


2 ( типичное значение;


3 ( максимальное значение;


4 ( расчет наихудшего случая (минимум/максимум).


Параметр IO_LEVEL указывает тип цифро-аналогового и аналого-цифрового интерфейса данного цифрового устройства: 


0 (  в соответствии со значением параметра DIGIOLVL директивы .OPTIONS (по умолчанию он равен 1);


1 (  интерфейс AtoD1/DtoA1;


2 (  интерфейс AtoD2/DtoA2; 


3 (  интерфейс AtoD3/DtoA3; 


4 (  интерфейс AtoD4/DtoA4. 


Запаздывание сигнала в примитивах цифровых устройств определяется в двух моделях: динамики и вход/выход. 


Модель динамики определяет задержки распространения и такие временные ограничения, как время установки (setup) и удерживания (hold). Модель вход/выход задает входные и выходные сопротивления и емкости и время переключения. 


Когда выход примитива соединяется с другим примитивом, общее время задержки распространения первого примитива равно сумме времени установления напряжения на его нагрузке и времени распространения сигнала, указанного в модели динамики. Время установления напряжения на нагрузке (loading delay) рассчитывается по формуле 





� EMBED Equation.2  ��� = 0,69R� EMBED Equation.2  ���C� EMBED Equation.2  ���, 





где R� EMBED Equation.2  ��� ( выходное сопротивление устройства, равное DRVH или DRVL в зависимости от логического уровня на выходе; C� EMBED Equation.2  ���  ( сумма входных и выходных емкостей цифровых устройств INLD, OUTLD, подключенных к данному выводу. 


Когда цифровой примитив подключен к аналоговому устройству, задержка распространения уменьшается на величину, равную времени переключения, заданного в модели вход/выход. 


Минимальная длительность сигнала на входе цифрового примитива, необходимая для изменения его логического состояния, должна превышать время задержки, приведенное в модели динамики (это ограничение не относится к цифровым линиям задержки). Более короткие входные импульсы не вызовут на выходе никакого эффекта. 


Приведем описание смешанной цепи на рис. 6.2, а: 





Analog/Digital Interface Example 


.OPT   ACCT   LIST   LIBRARY   EXPAND   RELTOL=.001 


.LIB   DIG.LIB 


VSIN   1   0   SIN(0 5v 1MEG) 


U1 STIM(1, 1) $G_DPWR $G_DGND 2 IO_STD TIMESTEP = 10ns 


+ (0C, 1) 


+ LABEL=BEGIN 


+ (1C,0) (2C,1) (3C,X) (4C,0) (5C,Z) 


+ 6C   GOTO   BEGIN   (1   TIMES 


X1   1   2   3   133LA3 


RL   3   0   25k 


CL   3   0   5pF 


.TRAN   5ns   500ns 


.PRINT   TRAN   V(1)   D(2)   V(3) 


.PROBE 


.END 





Здесь имеется обращение к библиотечному файлу моделей цифровых отечественных компонентов dig.lib, фрагменты которого приведены в Приложении 5. 


Обратим внимание, что цифровые ИС, даже простейшие, для которых имеются примитивы, представлены в библиотеке в виде макромоделей, имена которых совпадают с обозначением по ЕСКД (правда, в латинской транскрипции). Это позволяет пользователю не задумываться о правилах описания каждой конкретной ИС, предоставляя это разработчикам библиотек моделей. 


Информация об автоматически включаемых в схему макромоделях устройств сопряжения, их именах и именах новых цифровых узлов помещается в выходном файле с расширением .out: 





**** Generated AtoD and DtoA Interfaces **** 


* Analog/Digital interface for node 1 


* Moving X1.U1:IN1 from analog node 1 to new digital node 1$AtoD 


X$1_AtoD1 1 1$AtoD AtoD 


* Analog/Digital interface for node 3 


* Moving X1.U1:OUT1 from analog node 3 to new digital node 3$DtoA 


X$3_DtoA1 3$DtoA 3 DtoA 


* Analog/Digital interface power supply subckt 


X$DIGIFPWR 0 DIGIFPWR 





В качестве примера расчета переходных процессов приведем фрагмент выдачи данных в табличной форме по директиве .PRINT: 





TIME 		V(1) 			D(2) 	V(3) 


0.000E+00	0.000E+00 		1 	3.551E+00 


5.000E-09 	1.570E-01 		1 	3.551E+00 


1.000E-08 	3.139E-01 		0 	3.551E+00 


1.500E-08 	4.704E-01 		0 	3.551E+00 


2.000E-08 	6.264E-01 		1 	3.551E+00 


2.500E-08 	7.820E-01 		1 	3.551E+00 


3.000E-08 	9.369E-01 		X 	3.551E+00 


3.500E-08 	1.091E+00 		X	3.266E+00 





Перейдем теперь к описанию цифровых компонентов различных типов, сгруппировав их по следующим категориям: 


( многоразрядные АЦП и ЦАП; 


( вентили; 


( триггеры; 


( программируемые логические матрицы; 


( устройства памяти; 


( источники постоянных логических сигналов; 


( линии задержки;


( функциональное описание цифровых устройств. 

